
 **************************************************************************; 
* program:    Excise Taxes by Income Category                             ; 
* programmer: Rick Peterson                                               ; 
* project:    Washington Excise Tax Microsimulation Model                 ; 
* date:       April 27, 2002                                              ; 
*                                                                         ; 
* purpose:    Creates tables by income category of excise tax amouts      ; 
*             and tax as a percent of income. Incorporates data on        ; 
*             property tax and creates tables for total taxes             ; 
*                                                                         ; 
*-------------------------------------------------------------------------; 
* libraries: popsur - location of WA POP survey data                      ; 
*            Extaxmdl - location of microsimulation model data sets       ; 
*            hmdl2002 - location of Homeowner tax model                   ; 
*-------------------------------------------------------------------------; 
* incoming:   popsur.sps00f04 - WA POP survey                             ; 
*             Extaxmdl.taxbase1 - taxbase7 - Excise tax bases from        ; 
*             program called 'Create combined consumption tax base'       ; 
*             hmdl2002.HandRptx - Prop tax data from 2002 prop tax model  ; 
*-------------------------------------------------------------------------; 
* formats:   'Formats for excise tax microsimulation model 1.sas'         ; 
*            'Formats for excise tax microsimulation model 2.sas'         ; 
*-------------------------------------------------------------------------; 
* outgoing:   None                                          ; 
*                                                                         ; 
*-------------------------------------------------------------------------; 
* reports:   Tables 5, 6 9, and 10 for Excise tax microsimulation model   ; 
*            documentation                                                             
; 
*-------------------------------------------------------------------------; 
* changes:                                                                ; 
*                                                                         ; 
*-------------------------------------------------------------------------; 
* notes:                                                                  ; 
*                                                                         ; 
*                                                                         ; 
**************************************************************************; 
*-------------------------------------------------------------------------; 
*Get household data from WAPOP to merge with tax data; 
*-------------------------------------------------------------------------; 
Data z; 
set popsur.sps00f04;  
where pnum=1; 
keep id fnlwgt hhinc; 
run; 
proc sort data=z; 
by id; 
run; 
 
 
*-------------------------------------------------------------------------; 
*Apply State and Local tax rates to taxbases; 
*-------------------------------------------------------------------------; 
 
%macro loop1; 
%do x=1 %to 7; 
 
Data excisetaxes&x; 
set Extaxmdl.taxbase&x; 
*Sales tax; 

salesrate = .084; 
  SalesTax = salesbase*salesrate;   
*Alcohol taxes; 
 *Liquor sales and liter in container; 
 Liquorsalesrate_container = .205; 
 Liquorliterrate = 2.4408; 
  Liquorsalestax_container = 
liquorsalesbase_container*Liquorsalesrate_container; 
  Liquorvoltax_container = liquorvolbase_container * 
Liquorliterrate; 
   
 *Liquor sales and liter by the drink; 
 *Assume 50 percent markup after special sales tax is paid; 
 Liquorsalesrate_drink = .137; 
  Liquorsalestax_drink = 
.15*liquorsalesbase_drink*Liquorsalesrate_drink; 
  Liquorvoltax_drink = liquorvolbase_drink * Liquorliterrate; 
 
 *Wine tax; 
 Wineliterrate = .2292; 
  Winelitertax = Wineliterbase*wineliterrate; 
 
 *Beer Tax - rate per 31 gallons; 
 Beertaxrate = 8.08;; 
  Beertax = beerbase * beertaxrate; 
 
*Insurance Tax; 
 Insrate = .02; 
  InsTax = insrate*insbase; 
 
*Cigarette and other tobacco products taxes; 
 Cigrate = 1.425; 
  Cigtax = cigbase*cigrate; 
 Othertobrate = 1.294; 
  Othertobtax = othertobbase*othertobrate; 
 
*Public Utility Tax; 
 ElecPUTrate = .0378+.03201; 
  ElecPUT = ElecPUTrate*elecbase;  
 NatgasPUTrate = .03852+.03046; 
  NatgasPUT = NatgasPUTrate*naturalgasbase; 
 WaterseweragePUTrate = ((.05029+.03852)/2)+((.0535+.17192)/2); 
  WaterseweragePUT = WaterseweragePUTrate*waterseweragebase; 
 GarbagePUTrate = .036+.06372; 
  GarbagePUT = GarbagePUTrate*garbagebase; 
 IntercityPUTrate = .01926; 
  IntercityPUT = IntercityPUTrate*intercitybase; 
 IntracityPUTrate = .01926; 
  IntracityPUT = IntracityPUTrate*intracitybase; 
 
*Gas Tax; 
 Gasrate = .23; 
  Gastax = gasrate*gasbase; 
 
AlcoholTaxes = liquorsalestax_container + liquorvoltax_container + 
    liquorsalestax_drink + liquorvoltax_drink + 
Winelitertax + Beertax; 
TobaccoTaxes = cigtax + othertobtax; 
UtilityTaxes =  ElecPUT + NatgasPUT + WaterseweragePUT + GarbagePUT + 
    IntercityPUT + IntracityPUT; 

TotalExciseTaxes = Salestax + Alcoholtaxes + Instax + Tobaccotaxes +  
UtilityTaxes + Gastax; 
run; 
 
*Merge aggregated tax with household weight from WAPOP; 
 
data excisetaxmerged&x; 
merge excisetaxes&x(in=a) z; 
by id; 
if a; 
income=hhinc; 
households=1; 
drop _freq_ _type_; 
run; 
%end; 
%mend loop1; 
%loop1; 
*-------------------------------------------------------------------------; 
*Calculate total tax and tax as percent of income 
by income group for each imputation group; 
*-------------------------------------------------------------------------; 
 
*First, calculate tax by income group; 
 
OPTIONS ls=130 ps=50 pageno=1; 
%macro loop3; 
%do z=1 %to 7; 
proc summary nway data=excisetaxmerged&z; 
class hhinc ; 
format hhinc incfmtd.; 
weight fnlwgt; 
var TotalExciseTaxes Alcoholtaxes 
Tobaccotaxes UtilityTaxes salestax Instax Gastax households income; 
output out=weightedtaxbyinc&z sum=; 
run; 
 
*Divide by number of households by income group and calc avg tax; 
 
data avgtax; 
set weightedtaxbyinc&z; 
TotalExciseTaxes = TotalExciseTaxes/households; 
Salestax = salestax/households; 
Alcoholtaxes = Alcoholtaxes/households; 
Instax = instax/households; 
Tobaccotaxes = Tobaccotaxes/households; 
UtilityTaxes = UtilityTaxes/households; 
Gastax = gastax/households; 
drop _freq_ _type_; 
run; 
 
title1 "Excise Tax for Imputation Group&z" ; 
 
Proc tabulate data = avgtax; 
class hhinc; 
format hhinc incfmtd.; 
var TotalExciseTaxes Salestax  
Alcoholtaxes Instax Tobaccotaxes 
UtilityTaxes Gastax; 
table hhinc,(TotalExciseTaxes Salestax Alcoholtaxes Instax Tobaccotaxes 
UtilityTaxes Gastax)*f=comma10.; 

run; 
 
*Divide by total income by income group and calc tax as percent of income; 
 
data avgtaxpercent; 
set weightedtaxbyinc&z; 
TotalExciseTaxes = TotalExciseTaxes/income; 
Salestax = salestax/income; 
Alcoholtaxes = Alcoholtaxes/income; 
Instax = instax/income; 
Tobaccotaxes = Tobaccotaxes/income; 
UtilityTaxes = UtilityTaxes/income; 
Gastax = gastax/income; 
drop _freq_ _type_; 
run; 
 
title1 "Excise Tax as Percent of Income for Imputation Group&z" ; 
 
Proc tabulate data = avgtaxpercent ; 
class hhinc; 
format hhinc incfmtd.; 
var TotalExciseTaxes Salestax  
Alcoholtaxes Instax Tobaccotaxes 
UtilityTaxes Gastax; 
table hhinc,(TotalExciseTaxes  
Salestax Alcoholtaxes Instax Tobaccotaxes 
UtilityTaxes Gastax)*f=percent10.2; 
run; 
%end; 
%mend loop3; 
%loop3; 
 
*-------------------------------------------------------------------------; 
*Calculate Tax as a percent of income by income group for average of  
all seven imputation groups; 
*-------------------------------------------------------------------------; 
 
Data two; 
set excisetaxmerged1 excisetaxmerged2 excisetaxmerged3 excisetaxmerged4 
excisetaxmerged5 excisetaxmerged6 excisetaxmerged7; 
run; 
Proc summary data = two nway; 
class hhinc; 
format hhinc incfmtd.; 
var TotalExciseTaxes Alcoholtaxes  
Tobaccotaxes UtilityTaxes 
salestax Instax Gastax households income; 
weight fnlwgt; 
output out=Aveweightedtaxbyinc sum=; 
run; 
 
*Divide by total number of households by income group and calc avg tax; 
 
data avgtax1; 
set aveweightedtaxbyinc; 
TotalExciseTaxes = TotalExciseTaxes/households; 
salestax = salestax/households; 
Alcoholtaxes = Alcoholtaxes/households; 
Instax = instax/households; 
Tobaccotaxes = Tobaccotaxes/households; 

UtilityTaxes = UtilityTaxes/households; 
Gastax = gastax/households; 
drop _freq_ _type_; 
run; 
 
Title1 'Table 5'; 
title2 "State and Local Excise Tax by Income Category"; 
title3 "Average of all Imputation Groups" ; 
 
Proc tabulate data = avgtax1 ; 
class hhinc; 
format hhinc incfmtd.; 
var  TotalExciseTaxes  
salestax Alcoholtaxes Instax Tobaccotaxes 
 UtilityTaxes Gastax; 
table hhinc,(TotalExciseTaxes Salestax   
Alcoholtaxes Instax Tobaccotaxes 
UtilityTaxes Gastax)*f=comma10.; 
run; 
 
*Divide by total income by income group and calc tax as percent of income; 
 
data avgtax2; 
set aveweightedtaxbyinc; 
TotalExciseTaxes = TotalExciseTaxes/income; 
Salestax = salestax/income; 
Alcoholtaxes = Alcoholtaxes/income; 
Instax = instax/income; 
Tobaccotaxes = Tobaccotaxes/income; 
UtilityTaxes = UtilityTaxes/income; 
Gastax = gastax/income; 
drop _freq_ _type_; 
run; 
 
Title1 "Table 6"; 
title2 "State and Local Excise Tax as Percent of Income"; 
title3 "Average of all Imputation Groups" ; 
 
Proc tabulate data = avgtax2; 
class hhinc; 
format hhinc incfmtd.; 
var TotalExciseTaxes salestax  
Alcoholtaxes Instax Tobaccotaxes 
UtilityTaxes Gastax; 
table hhinc,(TotalExciseTaxes salestax  
Alcoholtaxes Instax Tobaccotaxes 
UtilityTaxes Gastax)*f=percent10.2; 
run; 
 
*-------------------------------------------------------------------------; 
*Create Tables with Excise and Property Taxes; 
*-------------------------------------------------------------------------; 
 
*Incorporate information on homeowner and renter prop tax; 
data x; 
set hmdl2002.HandRptx; 
propincome=hhinc; 
households=1; 
keep hhinc proptax propincome households fnlwgt; 
run; 

Proc summary nway data=x; 
class hhinc; 
format hhinc incfmtd.; 
var proptax propincome households; 
weight fnlwgt; 
output out=xx sum=; 
run; 
 
*Combine excise and property tax data for average tax table; 
 
data avgtax3; 
merge avgtax1(drop=households) xx; 
by hhinc; 
Proptax = proptax/households; 
TotalExciseandPropTaxes = TotalExciseTaxes + Proptax; 
run; 
 
Title1 'Table 9'; 
title2 "State and Local Excise and Property Tax as Percent of Income"; 
title3 "Average of all Imputation Group"; 
 
Proc tabulate data = avgtax3 ; 
class hhinc; 
format hhinc incfmtd.; 
var TotalExciseandPropTaxes TotalExciseTaxes Proptax; 
table hhinc,(TotalExciseandPropTaxes TotalExciseTaxes Proptax)*f=comma10.; 
run; 
 
*Combine excise and property tax data for tax as percent of income table; 
 
data avgtax4; 
merge avgtax2 xx; 
by hhinc; 
Proptax = proptax/propincome; 
TotalExciseandPropTaxes = TotalExciseTaxes + Proptax; 
run; 
 
Title1 "Table 10"; 
title2 "State and Local Excise and Property Tax as Percent of Income"; 
title3 "Average of all Imputation Group"; 
 
Proc tabulate data = avgtax4; 
class hhinc; 
format hhinc incfmtd.; 
var TotalExciseandPropTaxes TotalExciseTaxes Proptax; 
table hhinc,(TotalExciseandPropTaxes TotalExciseTaxes 
Proptax)*f=percent10.2; 
run; 
 




